Appendix A3. Partial Differential Equations

The wave equation

We show how Maple can be used to plot approximations to the solution of the wave equation on a finite domain. See Simmons/Krantz, Chapter 6, Section 2.

Consider string of length L along the x-axis with ends clamped at $x = 0$ and $x = L$. Let $y(x, t)$ denote the vertical displacement at time t of the point in string that is directly above (or below) the point $(x, 0)$. For small vibrations the function y satisfies the wave equation

$$y_{t,t} = a^2 y_{x,x}.$$

The letter a denotes a positive constant determined by the characteristics of the string. Separation of variables leads to solutions of the following form

$$Y_N(x, t) = \sum_{j=1}^{N} \left(a_j \cos\left(\frac{a\, j\, \pi\, t}{L}\right) + \mathrm{b}_j \sin\left(\frac{a\, j\, \pi\, t}{L}\right)\right) \sin\left(\frac{j\, \pi\, x}{L}\right), N \text{ a positive integer.}$$

Set the string into motion

The string is set into motion at $t = 0$ by giving it an initial shape $f(x)$ and an initial velocity distribution, $g(x)$. Thus the coefficients a_j and b_j should be chosen so that the function

$$Y_N(x, 0) = \sum_{j=1}^{N} a_j \sin\left(\frac{j\, \pi\, x}{L}\right)$$

approximates $f(x)$ on $[0, L]$ and the function

$$\left.\left(\frac{\partial}{\partial t} Y_N(x, t)\right)\right|_{t=0} = \sum_{j=1}^{N} \frac{a\, j\, \pi\, b_j}{L} \sin\left(\frac{j\, \pi\, x}{L}\right)$$

approximates $g(x)$. Consequently, a_j is the Fourier sine series coefficient for $f(x)$ and $\frac{a\, j\, \pi\, b_j}{L}$ is the Fourier sine series coefficient for $g(x)$.

The following entries define the functions f and g, calculate a_j and b_j, then create various solution curves. We assume that $L = 1$, $a = 1$ and the string is initially stretched "tent like" over the x-axis with shape

$$f(x) = piecewise(x < 0.5, 0.2\, x, 0.2\, (1 - x))$$

It is set into motion with a flick of the finger at a point one quarter of the way from the left endpoint

$$g(t) = 0.1\, \delta(t - 0.25)$$

You may, of course, change these to fit any situation that you would like to explore.

```
> L := 1: a := 1:
  f := x -> piecewise(x<L/2,2/5*x,2/5*(L-x)):
  g := x -> 1/10*Dirac(x - L/4):
  aj := 2/L*int(f(x)*sin(j*Pi*x/L), x=0..L):
  bj := L/(a*j*Pi)*2/L*int(g(x)*sin(j*Pi*x/L), x=0..L):
```

The following entry simplifies the formulas for aj and bj, then displays them.

```
> c := [aj,bj] assuming j::integer: 'aj'=c[1],'bj'=c[2];
```

$$aj = -\frac{2}{5}\,\frac{-2\sin\left(\frac{1}{2}\,j\,\pi\right)+\cos\left(\frac{1}{2}\,j\,\pi\right)j\,\pi}{j^2\,\pi^2}+\frac{2}{5}\,\frac{2\sin\left(\frac{1}{2}\,j\,\pi\right)+\cos\left(\frac{1}{2}\,j\,\pi\right)j\,\pi}{j^2\,\pi^2},\quad bj = \frac{1}{5}\,\frac{\sin\left(\frac{1}{4}\,j\,\pi\right)}{j\,\pi} \tag{1}$$

This it the definition of Y as a function of N, x, and t.

```
> Y := (N,x,t) ->
    sum((aj*cos(a*j*Pi*t/L)+bj*sin(a*j*Pi*t/L))*sin(j*Pi*x/L),
                         j=1..N);
```

$$Y := (N,x,t) \rightarrow \sum_{j=1}^{N}\left(aj\cos\left(\frac{a\,j\,\pi\,t}{L}\right)+bj\sin\left(\frac{a\,j\,\pi\,t}{L}\right)\right)\sin\left(\frac{j\,\pi\,x}{L}\right) \tag{2}$$

The first plot checks that the coefficients are correct for the velocity function g. A check for the shape function f is made when we plot Y at t =0 below.

```
> plots[setoptions]( tickmarks = [5,5] );
```

```
> plot( [g(x), sum(a*j*Pi*bj/L*sin(j*Pi*x/L), j=1..30)], x=0..L,
        color=[red,blue], title="Initial Velocity");
```



This approximation curve is a typical approximation to a Dirac delta. The area under the curve is approximately 1/10.

```
> evalf[4]( Int( sum(a*j*Pi*bj/L*sin(j*Pi*x/L), j=1..30), x=0..1));
```

$$0.1001 \tag{3}$$

The plot of Y(20,x,0) shows that the aj coefficients are also correct.

```
> plot( Y(20,x,0), x=0..L, 0..0.25, title="Initial Waveform");
```



Here is a snapshot of the waveform at $t = 0.05$:

```
> plot( Y(50,x,0.05), x=0..L, 0..0.25 );
```



Five snapshots, one every 0.2 seconds:

```
> plot( [Y(50,x,0.2*t)$t=0..5], x=0..L, -0.25..0.25 );
```



A movie (see the Help page for plots[animate]):

```
> plots[animate]( plot, [ Y(50,x,t), x=0..L , -0.25..0.25],
                  t=0..2, frames=40);
```



The waveform surface

```
> plot3d( Y(50,x,t), x=0..L, t=0..2,axes=boxed,orientation=[-50,70],
         lightmodel=light2 );
```